

Hi-Res Packing For the APPLE II

by R. M. Mottola
Cyborg Corp.
342 Western Ave.
Boston, MA 02135

It never fails. No matter what kind of program I write, if it makes use of high resolution graphics it's always too long. With HIRES screen #1 at \$2000 and screen #2 at \$4000, Applesoft Basic programs that make use of HIRES graphics must be either less than 6K in length or less than 14K in length. It must be some corollary of Murphy's Law that explains why the programs I write always come out just a few bytes too long. Ideally a 48K system with DOS should leave about 28K for your program and variables after 8K is allocated to the HIRES screen and 2K to system requirements. Why doesn't it work that way?

The main reason for this is that the memory allocated to the HIRES screen is located right smack in the middle of the area used to store Basic programs. Why they put them there is a mystery to me, although I might guess that when the hardware was designed (back in the olden days of expensive RAM and the idea that this little computer would be simply a "hobby" item) putting the screens on the 8K and 16K boundaries probably made a lot of sense.

In order to make full use of the available memory, it would be nice if we could somehow load our Applesoft Basic program "around" the HIRES screen we wish to use. The short machine language routine listed at the end of this article will do just that. Once BLOADED into page three of memory, you can modify any Applesoft program that's longer than 6K in length to include the HIRES screen #1 right in the program. The modified program will run, list, and otherwise behave perfectly normal. However, it will have a big hole in it, the HIRES screen, starting at \$2000 and ending at \$3FFF. As an added bonus, that hole will contain nothing but zeros. In effect, the HIRES screen will be pre-cleared so you will not have to explicitly clear it upon entering the HIRES mode the first time.

HOW TO LOAD AND USE IT

The easiest way to Type in the program is to put your Apple into the Monitor Mode with the asterisk (*) as the Prompt character. Then type the Machine Language directly into memory beginning at Address \$300 Hex (See page 44 of the Apple Reference Manual for directions). Your first entries should look like this:

```
*300:18 A5 B0 69 etc.
```

```
The Save on Disk using BSAVE (Program Name), A$300,L$AF
```

Use of the HIRES pack routine is very simple. First, BLOAD the routine at \$300. Then, LOAD into memory the Applesoft program you wish to modify. Once loaded, a CALL 768 will do the job of embedding the HIRES screen into your program. The whole process takes about a second. After that, SAVE your newly packed program onto disk. It's a very good idea to save it under a different name from the unmodified version of the same program. The reason for this will be explained in the section called "RESTRICTIONS".

HOW IT WORKS

In order to understand just how this utility works, it's essential to understand how an Applesoft program is stored in memory. The following discussion will attempt to describe this.

An Applesoft program is stored in memory as a singly linked list. Each line of the program contains in it an absolute pointer to the next line. These pointers are the first thing on each line. Therefore, the pointer on any line will point to the pointer of the next line, which points to the pointer of the next line... on and on to the end of the program. A block diagram of an Applesoft program line will look like this:

2 BYTE ABSOLUTE POINTER TO NEXT LINE	2 BYTE LINE NUMBER	TOKENIZED BASIC PROGRAM	END OF LINE CHARACTER (0)
--------------------------------------	--------------------	-------------------------	---------------------------

Since each line contains only one pointer which points to the next line in sequence, it can be seen that the only way to find something in a program is to start at the beginning and follow the chain of pointers through the program until you find what it is you are looking for, or until you run out of program — whichever comes first.


In our little routine, what we are looking for is the first line that lives at an address greater than \$1FFF or, in other words, is located in the same memory locations as the beginning of HIRES page #1. The routine just goes down the chain of pointers, looking for one that points past \$2000. Once found, it takes all of the rest of the program and moves it up to location \$4000, just past the end of the HIRES screen. Next it clears (zeros) all of the memory locations from \$2000 to \$3FFF. Finally, it changes the pointer of the line that pointed to the first line that was moved to \$4000, so that it does indeed point to the new position of that line. Now the list is again linked properly and will behave normally as a Basic program.

SOME RESTRICTIONS

The program listed will work just fine on an Apple II Plus or any Apple using Applesoft in ROM. It will not work for either Integer Basic programs or the RAM (cassette tape) version of Applesoft.

Once a program has been modified by this routine, almost any changes that you make to that program (DELETions, new lines, changed lines, etc.) will affect the positioning of the embedded HIRES screen. Although the program will look fine, as soon as it is run it will probably over-write portions of itself. You should never make changes to a program that has been modified by this utility. That's why it's a good idea to always keep two copies of the program on disk — one normal, the other modified with the embedded HIRES screen. The normal copy should not be run. Use it only as your source. That way, if your program does require some changes, you can make those changes to the normal version. Then, load and modify the newly changed version and save the it under a different file name. This will assure that you always have the ability to make program changes when required.

Remember that this routine will only be useful for programs longer than 6K in length. In fact, if you attempt to use it on a program that's



less than 6K in length, crazy things will happen to your system.

The routine uses HIMEM to determine if there is enough room in memory to include your program with the embedded HIRES screen. It will exit with an error if there isn't enough room. Therefore, it is important that you set HIMEM to the same value that it will be set to when the program is to be run. Do this before you attempt to modify any program with this utility. Remember, the modified program will be at least 8K longer than the unmodified version, because it will have the HIRES screen packed into it.

Whenever DOS does a LOAD, it does a little more than just load the program into memory from the disk. One of the extra things it does is re-link the list that the Basic code is stored as. Usually this has no consequence, but in this case it has the unfortunate result of un-linking all of the code on the high side of the HIRES screen. To prevent this, a special technique must be used to load in a program. First, BLOAD into page three (A\$300) the short routine named "LOAD PATCH" provided. A statement of the form:

```
100 PRINT D$; "BLOAD LOAD PATCH, A$300"
```

will do. Then perform the following POKES, to direct the DOS LOAD routine to the patch:

```
110 POKE 40288,0 : POKE 40289,3
```

The addresses provided in the above POKES are for a 48K system. If your system has less memory you must adjust them accordingly.

Once the POKES have been done, you are ready to LOAD and/or RUN your modified program. After the program is loaded into memory it's a good idea to reset the locations POKEd into above. You should have a line like:

```
10 POKE 40288,242 : POKE 40289,212
```

early in your program. This will return DOS to it's normal operation.

Next time you're writing a lengthy program that makes use of the Apple's HIRES graphics, don't worry if you're fast approaching the end-of-available-memory wall. Use this little utility and gain a lot of extra space for your program and variables.

```

0800 1 ;*****
0800 2 ;* *
0800 3 ;* APPLE II MACHINE *
0800 4 ;* LANGUAGE UTILITIES *
0800 5 ;* *
0800 6 ;* CYBORG CORPORATION *
0800 7 ;* BOSTON, MASS. *
0800 8 ;* *
0800 9 ;* R. M. MOTTOLA *
0800 10 ;* *
0800 11 ;*****

0800 14 ;*****
0800 15 ;* *
0800 16 ;* HIRES PACK ROUTINE *
0800 17 ;* *
0800 18 ;*****

0800 21 PT1L EPZ $18
0800 22 PT1H EPZ $19
0800 23 PT2L EPZ $1A
0800 24 PT2H EPZ $1B
0800 25 PT3L EPZ $1C
0800 26 PT3H EPZ $1D
0800 27 A1L EPZ $3C
0800 28 A1H EPZ $3D
0800 29 A2L EPZ $3E
0800 30 A2H EPZ $3F
0800 31 A4L EPZ $42
0800 32 A4H EPZ $43
0800 33 PROGL EPZ $67
0800 34 PROGH EPZ $68
0800 35 HIMEML EPZ $73
0800 36 HIMEMH EPZ $74
0800 37 ENDL EPZ $AF
0800 38 ENDH EPZ $B0
0800 39 NXTA4 EQU $FCB4
0800 40 MOVE EQU $FE2C
0800 41 PRERR EQU $FF2D
0300 43 ORG $300
0300 44 OBJ $800

0300 46 ; SEPARATES ANY APPLESOFT PROGRAM
0300 47 ; AND LEAVES SPACE IN THE MIDDLE
0300 48 ; OF IT FOR HIRES PAGE #1.

0300 18 50 SEPAR CLC
0301 A5B0 51 LDA ENDH
0303 6920 52 ADC #$20
0305 C574 53 CMP HIMEMH
0307 9003 54 BCC SEP1
0309 4C2DFF 55 JMP PRERR
030C A567 56 SEP1 LDA PROGL
030E 851C 57 STA PT3L
0310 A568 58 LDA PROGH
0312 851D 59 STA PT3H
0314 A51A 60 SEP2 LDA PT2L
0316 8518 61 STA PT1L
0318 A51B 62 LDA PT2H
031A 8519 63 STA PT1H
031C A51C 64 LDA PT3L
031E 851A 65 STA PT2L
0320 A51D 66 LDA PT3H
0322 851B 67 STA PT2H
0324 A000 68 LDY #$0
0326 B11C 69 LDA (PT3L),Y
0328 48 70 PHA
0329 C8 71 INY
032A B11C 72 LDA (PT3L),Y
032C 851D 73 STA PT3H
032E 68 74 PLA
032F 851C 75 STA PT3L
0331 A51D 76 LDA PT3H
0333 C920 77 CMP #$20
0335 90DD 78 BCC SEP2
0337 88 79 DEY
0338 38 80 SEC
0339 A900 81 LDA #$0
033B E51A 82 SBC PT2L
033D 851C 83 STA PT3L
033F A940 84 LDA #$40
0341 E51B 85 SBC PT2H
0343 851D 86 STA PT3H
0345 A900 87 LDA #$0
0347 9118 88 STA (PT1L),Y
0349 8542 89 STA A4L
034B C8 90 INY
034C A940 91 LDA #$40
034E 9118 92 STA (PT1L),Y
0350 8543 93 STA A4H
0352 88 94 DEY
0353 A51A 95 LDA PT2L
0355 853C 96 STA A1L
0357 A51B 97 LDA PT2H
0359 853D 98 STA A1H
035B A5AF 99 LDA ENDL
035D 853E 100 STA A2L
035F A5B0 101 LDA ENDH
0361 853F 102 STA A2H
0363 A000 103 LDY #$0
0365 202CFE 104 JSR MOVE
0368 98 105 TYA

```

```

0369 853C 106 STA A1L
036B A920 107 LDA #$20
036D 853D 108 STA A1H
036F A9FF 109 LDA #$FF
0371 853E 110 STA A2L
0373 A93F 111 LDA #$3F
0375 853F 112 STA A2H
0377 98 113 ERASE TYA
0378 913C 114 STA (A1L),Y
037A 20B4FC 115 JSR NXTA4
037D 90F8 116 BCC ERASE
037F 18 117 CLC
0380 A51C 118 LDA PT3L
0382 65AF 119 ADC ENDL
0384 85AF 120 STA ENDL
0386 A51D 121 LDA PT3H
0388 65B0 122 ADC ENDH
038A 85B0 123 STA ENDH
038C A900 124 LDA #$0
038E 851A 125 STA PT2L
0390 A940 126 LDA #$40
0392 851B 127 STA PT2H
0394 A000 128 SEP3 LDY #$0
0396 18 129 CLC
0397 B11A 130 LDA (PT2L),Y
0399 651C 131 ADC PT3L
039B 911A 132 STA (PT2L),Y
039D 48 133 PHA
039E C8 134 INY
039F B11A 135 LDA (PT2L),Y
03A1 651D 136 ADC PT3H
03A3 911A 137 STA (PT2L),Y
03A5 851B 138 STA PT2H
03A7 68 139 PLA
03A8 851A 140 STA PT2L
03AA B11A 141 LDA (PT2L),Y
03AC D0E6 142 BNE SEP3
03AE 60 143 RTS
144 END END

```

Hi-Res Packing (NIBBLE #4)

R.M. Mottola points out that the LOAD PATCH referred to in NIBBLE #4 (page 41) is not contained in the article. The code for this routine is listed below:

```

300: 20 65 06 18 A0 01 A5 69
308: 85 AF A5 6A 85 B0 4C 3C
310: 04

```

This code enters Basic without re-linking the Basic code. It allows programs with embedded Hi Res Screens to run normally. It is called from DOS 'LOAD' and is normally found at \$D4F2 in the Applesoft ROM.

R.M. also has indicated that the Hi Res Pack Routine has a limitation and will not work for programs longer than 12K in length. A memory list for a new routine (which removes this limitation) follows:

```

9513: A9 13 85 73 A9
9518: 95 85 74 18 A5 B0 69 20
9520: C5 74 90 03 4C 2D FF A5
9528: 67 85 1C A5 68 85 1D A5
9530: 1A 85 18 A5 1B 85 19 A5
9538: 1C 85 1A A5 1D 85 1B A0
9540: 00 B1 1C 48 C8 B1 1C 85
9548: 1D 68 85 1C A5 1D C9 20
9550: 90 DD 88 38 A9 00 E5 1A
9558: 85 1C A9 40 E5 1B 85 1D
9560: A9 00 91 18 85 42 C8 A9
9568: 40 91 18 85 43 88 A5 1A
9570: 85 3C A5 1B 85 3D A5 AF
9578: 85 3E A5 B0 85 3F A0 00
9580: 38 A5 3E E5 3C 48 A5 3F
9588: E5 3D 18 65 43 85 43 68
9590: 85 42 B1 3E 91 42 38 A5
9598: 42 E9 01 85 42 B0 02 C6
95A0: 43 38 A5 3E E9 01 85 3E
95A8: B0 02 C6 3F A5 3F C5 3D
95B0: B0 E0 A5 3E C5 3C B0 DA
95B8: 98 85 3C A9 20 85 3D A9
95C0: FF 85 3E A9 3F 85 3F 98
95C8: 91 3C 20 B4 FC 90 F8 18
95D0: A5 1C 65 AF 85 AF A5 1D
95D8: 65 B0 85 B0 A9 00 85 1A
95E0: A9 40 85 1B A0 00 18 B1
95E8: 1A 65 1C 91 1A 48 C8 B1
95F0: 1A 65 1D 91 1A 85 1B 68
95F8: 85 1A B1 1A D0 E6 60

```